

UTILIZAREA COMUNICATIEI PARALELE ÎN IBM – PC

1. Memento teoretic.

În PC comunicatia paralela se poate face prin intermediul a trei porturi de comunicatie, denumite LPT1, LPT2 si LPT3. Zona de date BIOS contine, începând de la adresa 0:0408h o lista a adreselor de baza corespunzatoare cuploarelor paralele. Adresele pentru LPT1 si LPT2 sunt:

LPT1 - 378h..37Fh

LPT2 - 278h..27Fh

Asupra acestor porturi se poate opera direct, prin instructiuni de intrare/iesire (IN, OUT), sau prin intermediul întreruperii BIOS 17h.

Utilizarea comunicatiei paralele se poate face prin interogare (*polling*), sau prin întreruperi. Liniile de întrerupere folosite, corespunzatoare controlorului de întreruperi 8259A, respectiv tipurile corespunzatoare ale întreruperilor în PC, sunt:

LPT1 - IRQ7 - INT 0Fh

LPT2 - IRQ5 - INT 0Dh

Registri asociati unui cuplor de comunicatie paralela (adresele fiind date pentru LPT1), sunt:

Adresa

378h Registrul de date al cuplorului paralel.

Scriere: contine caracterul ASCII trimis imprimantei.

Citire: Contine ultimul caracter trimis.

37Ah *Citire/scriere*: Registrul de control al imprimantei.

Bit: 0 - strob, este 1 la transmiterea unui octet;

1 - AUTO LineFeed, 1 cauzeaza trimiterea automata a unui LF dupa un CR;

2 - INIT, initializare, 0 reseteaza imprimanta;

3 - SLCT IN, selectare, 1 selecteaza imprimanta;

4 - IRQ Enable, validare întreruperi, 1 - permite generarea unei întreruperi *hardware* când ACK este 0;

5, 6, 7 - sunt 0.

379h *Citire*: Registrul de stare al imprimantei.

Bit: 0,1,2 - sunt 0

3 - ERROR, cand e 0 imprimanta semnalizeaza o eroare;

4 - SLCT, 1: imprimanta e selectata;

5 - PE, 1: lipsa hârtie în imprimanta;

6 - ACK, 0: este permisa trimiterea urmatorului caracter;

7 - BUSY, 0: *buffer*-ul imprimantei este plin, imprimanta este decuplata, sau a aparut o alta eroare;

Întreruperea BIOS 17h pune la dispozitie urmatoarele servicii pentru interfata paralela:

AH=00h - Imprimarea unui caracter.

La apel:

AL - codul ASCII al caracterului;

DX - numarul imprimantei (0, 1 sau 2);

La revenire:

AH - starea imprimantei, semnificatiile bitilor fiind date la functia 02h;

AH=01h - Initializarea portului paralel.

La apel:

DX - numarul imprimantei;

La revenire:

AH - starea imprimantei, semnificatiile bitilor fiind date la functia 02h;

AH=02h - Citirea starii imprimantei.

La apel:

DX - numarul imprimantei.

La revenire:

AH - starea imprimantei (returnata si de functiile 0 si 1)

bit 0 - 1: *timeout*;

bit 1, 2 - nefolositi;

bit 3 - 1: eroare I/O;

bit 4 - 0: imprimanta decuplata, 1: imprimanta selectata;

bit 5 - 1: lipsa hârtie în imprimanta;

bit 6 - 1: ACK;

bit 7 - 1: imprimanta *ready*.

2. Programe demonstrative.

Exemplul 1: Imprimarea unui sir de caractere prin interogare (*polling*).

```

.model small

buff equ 278h           ;registrul date LPT2
stat equ 279h           ;registrul stare LPT2
cont equ 27Ah           ;registrul control LPT2
ack equ 01000000b       ;bitul ACK

.data

text db 'Test imprimanta',13,10
lung equ $-text         ;lungimea textului

.code

start: mov ax,@data      ;încarcarea registrului ds
        mov ds,ax        ;cu baza segmentului de date
        mov dx,cont      ;registrul de control
        mov al,08h       ;selectare imprimanta
        out dx,al
        mov si,offset text
        mov cx,lung       ;lungimea textului
et1:    lodsb            ;al <- ds:[si], si++
        mov dx,buff       ;registrul de date
        out dx,al        ;imprimare caracter
        mov dx,cont      ;registrul de control
        mov al,0Fh        ;strob = 1
        out dx,al
        mov al,0Eh       ;strob = 0
        out dx,al

```

```

        mov dx,stat                ;registrul de stare
et2:    in al,dx                   ;citire stare
        test al,ack                ;gata pentru urmatorul caracter?
        jnz et2
        loop et1
        mov ax,4C00h              ;iesire în DOS
        int 21h
end start

```

Exemplul 2: Imprimare prin folosirea întreruperilor.

```

.model small

.stack 100h

.data

int0foff    dw 0                  ;deplasamentul "vechii" rutine de tratare
int0fseg    dw 0                  ;segmentul "vechii" rutine de tratare
mes         db 13,10,'Pentru terminare apasati o tasta!$'
oldmask     db 0                  ;masca întreruperilor în controlorul 8259A

.code

contor      db 10                 ;contorul caracterelor imprimate pe o linie

;Rutina de întârziere
del:        push cx
            mov cx,0FFh
e:          dec cx
            or cx,cx

```

```
jnz e
pop cx
ret
```

;Rutina de imprimare a unui caracter

printch:

```
push ax                ;salvarea registrilor folositi în rutina
push dx
mov dx,378h
out dx,al              ;imprimarea caracterului
mov al,0FFh            ;strob on
mov dx,37Ah
out dx,al
mov al,0FEh            ;strob off
out dx,al
pop dx                 ;refacerea registrilor la valorile anterioare
pop ax
ret
```

;Rutina de tratare a întreruperii LPT1 (int 0Fh)

new_int0f:

```
push ax                ;salvarea registrilor folositi în rutina
push dx
mov al,'A'             ;imprimarea a 10 caractere 'A' pe o linie
dec cs:contor
jnz et1
mov al,10              ;daca s-a ajuns la al 10-lea
mov cs:contor,al       ;initializare contor
```

mov al,13	;imprimare CR
call printch	
mov al,10	;imprimare LF
et1: call printch	
mov al,20h	;EOI pentru 8259A
out 20h,al	
pop dx	;refacerea registrilor implicati în rutina
pop ax	
iret	
;Programul principal	
start: mov ax,@data	;initializarea registrului DS
mov ds,ax	;cu baza segmentului de date
mov ax,350Fh	;citirea vectorului întreruperii 0Fh
int 21h	
mov int0foff,bx	;salvarea offset-ului
mov int0fseg,es	;salvarea segmentului
mov ax,250Fh	;capturarea întreruperii 0Fh
push ds	;salvarea ds
push cs	
pop ds	;ds=cs
mov dx,offset new_int0f	;offset-ul rutinei proprii
int 21h	
pop ds	;refacerea registrului ds
in al,21h	;citirea mastii din registrul 8259A
mov oldmask,al	;salvarea mastii

and al,7Fh	;demascarea întreruperii LPT1 (0Fh)
out 21h,al	;setarea noii masti
mov dx,37Ah	;scrierea registrului de control cu INIT=0
mov al,1Ah	
out dx,al	
call del	;asteptare
mov al,1Eh	;scrierea registrului de control cu INIT=1
out dx,al	
mov ah,09h	;afisare mesaj
mov dx,offset mesaj	
int 21h	
mov ah,1	;asteptare o tasta
int 21h	
mov al,oldmask	
out 21h,al	;refacerea vechii masti
mov dx,37Ah	
mov al,0Ah	;invalidare întrerupere LPT1+ INIT=0
out dx,al	
call del	;asteptare
mov al,0Eh	;INIT=1
out dx,al	
mov ax,250Fh	;refacerea vechiului vector de întrerupere
mov dx,int0fseg	
mov ds,dx	


```
mov dx,int0foff
int 21h

mov ax,4C00h                ;revenirea în DOS
int 21h

end start
```

3. Probleme propuse.

1. Sa se încerce programele de mai sus, adaugându-se tratarea erorilor ce pot apărea (imprimanta decuplata, lipsa hârtie, etc.).

2. Sa se scrie un program de testare a registrilor cuploarelor LPT1 si LPT2, prin scrierea a câte unui bit pe 0, respectiv 1, urmata de verificarea bitului respectiv prin citire si comparare. În cazul aparitiei unei erori sa se afiseze un mesaj corespunzator.

3. Sa se scrie un program rezident care indirecteaza întreruperea serviciilor pentru imprimanta (int 17h), cu scopul salvarii într-un fisier a tuturor datelor trimise imprimantei.